PROGRAM : INTEGRAL by Jan Zijlstra, Middle Tennessee State University.

PURPOSE : To estimate the area bounded by the graph of a the function *f*(*x*) and the *x*-axis.

PLATFORM: Texas Instruments TI 83/84 graphing calculator

The following approximations of the area under a curve on the interval [*a*,*b*] are computed:

- *Left* (-hand) *Sum* : $L_n = \sum_{i=1}^{n} f(x_{i-1})\Delta x_i$
- *Right* (-hand) *Sum*: $R_n = \sum_{i=1}^{n} f(x_i)\Delta x_i$
- *Trapezoid Rule* : $T_n = \sum_{i=1}^{n} \frac{f(x_{i-1})+f(x_i)}{2}\Delta x_i = \frac{L_n + R_n}{2}$
- *Midpoint Rule* : $M_n = \sum_{i=1}^{n} f\left(\frac{x_{i-1}+x_i}{2}\right)\Delta x_i$
- *Simpson's Rule* : $S_n = \frac{2M_n + T_n}{3}$

Input : The function *f*(*x*) is stored in $\mathbf{y}_1$ prior to running the program. The limits of integration (*a* and *b*) and the number of subdivisions *n* are prompted for by the program.

Output: The estimates described above, stored in *L*, *R*, *T*, *M* and *S*.

Note: For the program to successfully estimate the area under a curve make sure that the area is *bounded* on the interval [a, b] (no vertical asymptotes)

```
 PROGRAM INTEGRAL                  Command Locations
:ClrHome                           ClrHome: Catalog
:Prompt A,B,N                      Prompt: prgm I/O
:(B-A)/N → H                       →: Sto
:0 → R
:0 → M
:For(I,1,N,1)                      For: Prgm Ctl
:R+y1(A+IH)*H →R                   y1: VARS YVARS Function 1
:M+y1(A+(I-1/2)*H)*H →M
:End                               End: Prgm Ctl
:R+H*(y1(A)-y1(B))→L
:(L+R)/2 →T
:(2M+T)/3 →S
:ClrHome                           ClrHome: Catalog
:Output(1,1,"L:")                  Output: Prgm I/O 6
:Output(1,3,L)
:Output(2,1,"R:")
:Output(2,3,R)
:Output(3,1,"T:")
:Output(3,3,T)
:Output(4,1,"M:")
:Output(4,3,M)
:Output(5,1,"S:")
:Output(5,3,S)
```

EXAMPLE: Using program INTEGRAL, we can approximate the area under the curve $f(x) = \frac{1}{x}$ on the interval [1,3] with $n = 4$ subintervals as:

Method	Estimate	Relative Error (%):
Left Sum	1.28333	16.8
Right Sum	0.95000	-13.5
Trapezoid Rule	1.11667	1.64
Midpoint Rule	1.08975	-.81
Simpson's Rule	1.09872	0.01

The relative error is expressed in percent of the true value $\int_1^3 1/x \, dx = \ln 3$

EXAMPLE: Using program INTEGRAL, we can approximate the area under the curve $f(x) = \sqrt{x}$ on the interval [0,1] with $n = 4$ subintervals as:

Method	Estimate	Relative Error (%):
Left Sum	0.518283	-22.2
Right Sum	0.768283	15.2
Trapezoid Rule	0.643283	-3.51
Midpoint Rule	0.672977	0.95
Simpson's Rule	0.663079	-0.54

The relative error is expressed in percent of the true value $\int_0^1 \sqrt{x} \, dx = 2/3$

EXERCISES:

1. Estimate the area bounded by the graph of $f(x) = \ln(x)$ on the interval [1,3] Using the trapezoid rule with 6 subintervals. (Answer: $T = 1.289696$)

2. Estimate the area bounded by the graph of $f(x) = \sin(x)$ on the interval $[0, \pi]$ Using Simpson's rule with 5 subintervals. (Answer: $T = 2.000109$)
   Estimate the area bounded by the graph of $f(x) = \sin(x)$ on the interval $[0, 2\pi]$ Using Simpson's rule with 10 subintervals. (Answer: $T = 0$). *Why* is the area zero?